RSW 9-11 bank tube test
8 mode
start 20

RSW checked only at begin
(every 6 minutes)

wipes out RIM & BIN

IDENTIFICATION

Product Code:	MAINDEC -12-D1BA
Product Name:	JMP SELF
Date Created:	September 12, 1969
Maintainer:	Diagnostic Group
Author:	James Kelly

1. ABSTRACT

The JMP Self test is a worst case test of the core memory Read/Write gates. The program loads all of core memory from address 0240 to 7777 inclusive in bank 0 and the entire memory bank for extended memories to (JMP Self).

The program types a blank character on the teletype, turns on program interrupt and jumps to the memory location to be tested in either memory bank 0 or the selected extended memory. When the program interrupt occurs, a test is made to be sure that we interrupted from the correct memory location. Any errors will be indicated by an error halt and a message typeout, depending on the switch settings.

2. MACHINE REQUIREMENTS

a. A standard PDP-5, 8, 8/S, 8I, 8L, 12 or Linc-8.
b. An ASR-33 teletype or equivalent.
c. If the PDP-5 being tested has extended memory, the CIF and CDF instructions must be compatible with the PDP-8.

2.2 Preliminary Programs

All basic instruction and memory diagnostics must have been successfully run prior to attempting to run (JMP Self).

3. LOADING PROCEDURES

3.1 Method

This program must be loaded with the binary loader. If you are unfamiliar with the proper binary loading procedures, refer to the User Handbook for your computer.

a. Set the teletype reader switch to FREE.
b. Open the teletype reader and insert the program tape so that the arrows on the tape are visible to, and pointing toward the operator.
c. Close the reader and set the reader switch to START.
d. Set the teletype front panel switch to ON-LINE.
e. Set the LEFT switches to 7777.
f. Set the RIGHT switches to 4000.
g. Set the MODE switch to 8 mode.
h. Depress I/O preset.
i. Depress START LS.
j. When the program tape has been read in, the computer will halt.
k. The ACCUMULATOR must be equal to 0000; if it is not, an error has occurred and one might try reloading the binary loader.

4. STARTING PROCEDURE

a. Remove the paper tape from the teletype reader.
b. Set the three right most switches SR9, 10, 11 to the number of the memory bank you wish to test. In a basic machine with no extended memory, this would be 000.
c. Set the MODE switch to 8 mode.
d. Depress I/O preset.
e. Depress START 20.
f. The program, when properly running, will cause the PROGRAM COUNTER and MEMORY ADDRESS register to appear to be counting up, and the ION indicator will light.
g. NOTE: Attempting to test extended memory in a 4K machine will over-write the diagnostic and destroy the program.

4.1 Switch Settings

In general, switches 0, 1, 2 allow the test engineer to select the mode of error indication, i.e. type out or error halt. The normal mode with switches 0, 1, 2 on a zero is an error halt. To modify these circumstances procede as follows:

SR00 = 1 Suppress halt; depress continue for printout or loop
SR01 = 1 Suppress typing
SR02 = 1 Scope Loop on error

These designated switches have an order of precedence associated with them, which is designed for maximum flexibility.

In the event of an error, the first switch to be tested is switch 0; if it is 0 the computer will halt at address 0063. If it is a 1, i.e. suppress halt, we test switch 1. If switch 1 is 0 the following "typical" error message will ensue:

```
JMP.
GOOD   BAD   ADDR
 0377   0357  5357
```

This message is interpreted as follows:

1) The "GOOD" address from which the program interrupt should have occurred. In other words, the address of the (JMP.) we were supposed to be performing.

2) The "BAD" address from which the program interrupt actually occurred.

3) The "ADDR" number refers to the contents of the "good" or memory location under test. In this case it can be seen that bit 07 of the (JMP.) instruction was dropped causing the computer to Jump Not to itself in 0377, but rather to 0357.

4) In some cases the number under BAD will be the address GOOD +1. This usually indicates that bit Ø2 was dropped changing the JMP self to JMS self and inserting the current address +1 into the current location.

Placing the RIM loader in core memory by way of the operator console keys and switches is accomplished as follows:

a. Set the starting address 7756 in the LEFT switches.
b. Set the first instruction (6032) in the RIGHT switches.
c. Press the FILL switch, then press FILL STEP.
d. Set the next instruction (6031) in the RIGHT switches.
e. Press the FILL STEP switch.
f. Repeat steps d and e until all 16 instructions have been deposited.

To read a tape in RIM format, place the tape in the reader, set the LEFT switches to the starting address 7756 of the RIM loader (not of the program being read), press the START LS key, and start the Teletype reader.

BINARY FORMAT PERFORATED TAPE LOADER

Once the RIM loader is in core, place the binary loader program tape on the teletype reader and turn the reader on. Set the LEFT switches to 7756, depress I/O preset with the mode switch in 8 mode, then depress START LS. The binary tape will read into core. The reader must be turned off manually as the tape reaches the end, since RIM does not stop.

APPENDIX A

PDP-8 MODE PERFORATED-TAPE LOADER

READIN MODE LOADER

The readin mode (RIM) loader is a minimum length, basic, perforated-tape program for the 33 ASR. It is initially stored in memory by manual use of the operator console keys and switches. The loader is permanently stored in 18 locations of page 37.

The RIM loader can only be used in conjunction with the 33ASR reader (not the high-speed perforated-tape reader). Because a tape in RIM format is, in effect, twice as long as it need be, it is suggested that the RIM loader be used only to read the binary loader when using the 33 ASR. (NOTE: Some PDP-12 diagnostic program tapes are in RIM format).

The complete PDP-12 RIM loader (SA=7756) is as follows:

Absolute Address	Octal Content	Tag	Instruction I Z	Comments
7756	6032	BEG,	KCC	/CLEAR AC AND FLAG
7757	6031		KSF	/SKIP IF FLAG=1
7760	5357		JMP-1	/LOOKING FOR CHARACTE
7761,	6036		KRB	/READ BUFFER
7762,	7106		CLL RTL	
7763,	7006		RTL	/CHANNEL 8 IN ACO
7764,	7510		SPA	/CHECKING FOR LEADER
7765,	5357		JMP BEG+1	/FOUND LEADER
7766,	7006		RTL	/OK, CHANNEL 7 IN LINK
7767,	6031		KSF	
7770,	5367		JMP-1	
7771,	6034		KRS	/READ, DO NOT CLEAR
7772,	7420		SNL	/CHECKING FOR ADDRESS
7773,	3776		DCA 1 TEMP	/STORE CONTENT
7774,	3376		DCA TEMP	/STORE ADDRESS
7775,	5356		JMP BEG	/NEXT WORD
7776,	0	TEMP,	0	/TEMP STORAGE
7777	5XXX		JMP X	/JMP START OF BIN LOADER

```
/JMP SELF PDP-12
/COPYRIGHT 1969, DIGITAL EQUIPMENT CORP., MAYNARD, MASS.
/AUTHOR: JAMES KELLY
/
/SR00=1 INHIBIT ERROR HALT
/SR01=1 INHIBIT ERROR TYPE OUT
/SR02=1 SCOPE LOOP ON ERROR

/THIS TEST IS DESIGNED TO TEST THE ABILITY
/OF THE MEMORY ADDRESS SELECT GATES TO
/SWITCH RAPIDLY BETWEEN READ CURRENT AND
/WRITE CURRENT I.E. REVERSE DIRECTION.

/MEMORY LOCATIONS 0240 THRU 7777 IN MEMORY
/BANK 0 AND ALL MEMORY LOCATIONS IN THE EXTENDED
/MEMORY BANKS ARE LOADED TO A (JMP.) CONDITION.
/THE TELETYPE PRINTER FLAG IS CLEARED AND A BLANK
/CHARACTER IS TYPED OUT. IMMEDIATELY THE INTERRUPT
/IS TURNED ON AND A JUMP TO THE (JMP.) LOCATION
/IS EXECUTED, UPON COMPLETION OF THE TELEPRINTER
/OPERATION THE COMPUTER INTERRUPTS AND A TEST
/IS MADE TO BE SURE WE INTERRUPTED FROM THE
/CORRECT MEMORY ADDRESS. IF NO ERRORS OCCURRED
/THE PROGRAM PROCEEDES TO TEST EACH MEMORY
/LOCATION IN THE SELECTED BANK.

/TO TEST ANY AMOUNT OF EXTENDED MEMORY
/SET SWITCHES 9,10,11 TO THE BANK TO BE TESTED
/AND START THE PROGRAM AT THE BEGINNING.

/SR09=EXTENDED MEMORY
/SR10=EXTENDED MEMORY
/SR11=EXTENDED MEMORY
/ATTEMPTING TO TEST NON-EXISTANT MEMORY WILL
/RESULT IN FALSE ERROR PRINTOUTS OR PROGRAM DESTRUCTION
```

```
                    /
                    /THE FOLLOWING INTERRUPT ROUTINES WORKS ON PDP-5/12
                    /
          0001      *1

0001      0000      INTDTA, 0000                        /PDP-5 INTERRUPT
0002      1201              TAD     INTDTA              /GET PDP-5 INTERRUPT
0003      7450              SNA                         /IF AC=0 WERE IN AN 8 OR 12
0004      1200              TAD     INTDTA+1            /GET PDP-8 INTERRUPT
0005      3207              DCA     INTSTO              /STORE FOR TYPING
0006      5257              JMP     PNTA                /EXIT TO TEST ROUTINE
0007      0000      INTSTO, 0000                        /INTERRUPT STORAGE DATA

          0010      *10

0010      0000      AUTO10, 0000                        /TYPE OUT POINTER
0011      0070      K0070,  0070
0012      0000      TEMP,   0000
0013      6202      K6202,  6202
0014      0177      K0177,  0177
0015      5200      K5200,  5200
0016      0240      K0240,  0240
0017      7774      K7774,  7774

                    /
                    /DETERMINE MEMORY FIELD
                    /

          0020      *20

0020      7604      START,  LAS                         /GET BANK DATA
0021      7006              RTL                         /MOVE SR09,10,11
0022      7004              RAL
0023      0211              AND     K0070               /SAVE FIELD DATA
0024      3207              DCA     INTSTO              /STORE FIELD DATA
0025      1207              TAD     INTSTO              /FETCH IT
0026      1076              TAD     K6201               /CHANGE DATA FIELD
0027      3237              DCA     BEGIN
0030      1207              TAD     INTSTO              /GET FIELD DATA
0031      1213              TAD     K6202               /CHANGE INST FIELD
0032      3253              DCA     CIFLOC+1            /STORE
                    /
                    /DETERMINE LOWER LIMIT OF TEST
                    /
0033      1207              TAD     INTSTO              /GET CHANGE FIELD DATA
0034      7650              SNA CLA                     /FIELD 0 OR EXTENDED
0035      1216              TAD     K0240               /FIELD 0
0036      3212              DCA     TEMP                /STORE EITHER 0000 OR 0240
                    /
                    /LOAD SELECTED MEMORY BANK WITH (JMP DOT)
                    /
0037      0000      BEGIN,  0000                        /CHANGE DATA FIELD
0040      1212              TAD     TEMP                /GET LOWER LIMIT
```

```
0041  3152          DCA     TALLY           /SET TALLY
0042  1152          TAD     TALLY           /GET IT
```

```
0043  0014          AND     K0177           /SAVE RELATIVE ADDRESS
0044  1015          TAD     K5200           /ADD BASIC JMP.
0045  3552          DCA I   TALLY           /STORE IT
0046  2152          ISZ     TALLY           /UPDATE POINTER
0047  5042          JMP     BEGIN+3         /DO SOME MORE
0050  1012          TAD     TEMP            /GET POINTR
0051  3152          DCA     TALLY           /RESET TALLY
                    /
                    /GO TO TEST LOCATION
                    /
0052  6046  CIFLOC, TLS                     /HIT TELETYPE
0053  0000          0000                    /CHANGE INSTRUCTION FIELD
0054  6032          KCC                     /CLEAR TELEPRINTER FLAG
0055  6001          ION                     /TURN ON INTERRUPT
0056  5552          JMP I   TALLY           /GO TO JMP.
                    /
                    /TEST ROUTINE
                    /
0057  6031  PNTA,   KSF                     /FALSE INTERRUPT?
0060  5062          JMP     .+2             /NO
0061  5053          JMP     CIFLOC+1        /YES, GO BACK
0062  1007          TAD     INTSTO          /GET INT DATA
0063  7041          CIA                     /NEGATE
0064  1152          TAD     TALLY           /SUBTRACT TALLY
0065  7640          SZA CLA                 /TEST
0066  5555          JMP I   GOOF            /GOOF
0067  2152          ISZ     TALLY           /UPDATE (JMP DOT) POINTER
0070  5052          JMP     CIFLOC          /DO AGAIN
0071  5020          JMP     START           /START OVER
```

```
                        /
                        /TYPE OUT ROUTINE
                        /
0072  7300    TALK,     CLA CLL                 /CLEAR ACL
0073  1010              TAD     AUTO10          /GET AUTO10=0000 NEVER TYPED
0074  7640              SZA CLA                 /=0?
0075  5106              JMP     DATA            /NO TYPE NUMERICS
0076  6201    K6201,    6201                    /RESTORE DATA FIELD 0
0077  1153              TAD     MESSA           /GET POINTER
0100  3010              DCA     AUTO10          /STORE IN AUTO10
0101  1410              TAD I   AUTO10          /FETCH A CHARACTER
0102  7450              SNA                     /DONE YET
0103  5106              JMP     .+3             /YES
0104  4554              JMS I   TYPE            /TYPE IT
0105  5101              JMP     .-4             /NO
                        /
                        /DATA TYPE OUT
                        /
0106  1152    DATA,     TAD     TALLY           /GET ADDRESS
0107  4126              JMS     OCTYP           /TYPE
0110  1007              TAD     INTSTO          /GET ERROR
0111  4126              JMS     OCTYP
0112  1037              TAD     BEGIN           /GET BANK
0113  3114              DCA     .+1             /STORE IT
0114  0000              0000                    /CHANGE BANKS
0115  1552              TAD I   TALLY
0116  4126              JMS     OCTYP
0117  1160              TAD     K215
0120  4554              JMS I   TYPE
0121  1161              TAD     K212
0122  4554              JMS I   TYPE
0123  7604              LAS
0124  7006              RTL
0125  5556              JMP I   PNTB
0126  0000    OCTYP,    0
0127  3012              DCA     TEMP
0130  1017              TAD     K7774
0131  3157              DCA     CNTR
0132  1151    HERE,     TAD     K1026
0133  3010    REDO,     DCA     AUTO10
0134  1012              TAD     TEMP
0135  7004              RAL
0136  3012              DCA     TEMP
0137  1010              TAD     AUTO10
0140  7004              RAL
0141  7420              SNL
0142  5133              JMP     REDO
0143  4554              JMS I   TYPE
0144  2157              ISZ     CNTR
0145  5132              JMP     HERE
0146  1016              TAD     K0240
0147  4554              JMS I   TYPE
0150  5526              JMP I   OCTYP
```

```
0151  1026    K1026,  1026
0152  0000    TALLY,  0000
0153  0153    MESSA,  MESSA
0154  0211    TYPE,   TYPOUT
0155  0217    GOOF,   GOOFN
0156  0226    PNTB,   PNTBN
0157  0000    CNTR,   0

0160  0212    K215,   0212                    /CR
0161  0215    K212,   0215                    /LF
0162  0312            0312                    /J
0163  0315            0315                    /M
0164  0320            0320                    /P
0165  0256            0256                    /.
0166  0215            0215                    /CR
0167  0212            0212                    /LF
0170  0307            0307                    /G
0171  0317            0317                    /O
0172  0317            0317                    /O
0173  0304            0304                    /D
0174  0240            0240                    /SPACE
0175  0302            0302                    /B
0176  0301            0301                    /A
0177  0304            0304                    /D
0200  0240            0240                    /SPACE
0201  0240            0240                    /SPACE
0202  0301            0301                    /A
0203  0304            0304                    /D
0204  0344            0344                    /D
0205  0322            0322                    /R
0206  0215            0215                    /CR
0207  0212            0212                    /LF
0210  0000            0000

0211  0000    TYPOUT, 0
0212  6046            TLS
0213  6041            TSF
0214  5213            JMP     .-1
0215  7200            CLA
0216  5611            JMP I   TYPOUT
                      /
                      /ERROR HANDLER
                      /
0217  7604    GOOFN,  LAS                     /READ SWITCHES
0220  7500            SMA                     /SR00=?
0221  7402            HLT                     /ERROR HALT
0222  7004            RAL                     /MOVE SR01 TO AC0
0223  7500            SMA                     /SR01=?
0224  5072            JMP     TALK            /TYPE
0225  7004            RAL                     /MOVE SR02 TO AC0
0226  7700    PNTBN,  SMA CLA                 /SR02=?
0227  5067            JMP     PNTA+10         /GO
0230  1037            TAD     BEGIN           /GET EXT MEM
0231  3232            DCA     .+1             /CHANGE DATA FIELD
0232  0000            0000                    /SET DATA FIELD
```

```
0233  1152          TAD     TALLY           /GET ADDRESS
0234  0014          AND     K0177           /SAVE RELATIVE ADDRESS
0235  1015          TAD     K5200           /AND BASIC JUMP
0236  3552          DCA I   TALLY           /STORE IT
0237  5052          JMP     CIFLOC          /GO TO TEST
                    $
```